

גירסה 2.00 - 19.3.2002

ASP – Part 2

מסמך זה הורד מהאתר <http://underwar.livedns.co.il> אין להפיץ מסמך זה במדיה כלשהי, ללא אישור מפורש מאת המחבר. מחבר המסמך איננו אחראי לכל נזק, ישיר או עקיף, שיגרם עקב השימוש במידע המופיע במסמך, וכן לנכונות התוכן של הנושאים המופיעים במסמך. עם זאת, המחבר עשה את מירב המאמצים כדי לספק את המידע המדויק והמלא ביותר.

כל הזכויות שמורות לניר אדר

Nir Adar

Email: underwar@hotmail.com

Home Page: <http://underwar.livedns.co.il>

מסמך זה מהווה המשך למסמך הראשון בסדרה זו. ההנחה היא שהקורא שולט בחומר המופיע במסמך הראשון. במסמך זה נמשיך להסביר על ASP. הנושאים המופיעים במסמך: JSP, אובייקטים בASP והגבה לטפסים (FORMS).

JSP

ניתן לכתוב דפי ASP המבוססים על JScript.
על מנת לעשות זאת, מוסיפים את השורה:

```
<%@ LANGUAGE=JScript %>
```

בראש הדף.

מגדירים משתנים בJScript בעזרת הפקודה var, לדוגמא:

```
var age
```

הנה דוגמא לדף פשוט מבוסס JScript:

```
<%@ LANGUAGE=JScript %>
```

```
<html>
```

```
<head>
```

```
<title></title>
```

```
</head>
```

```
<body>
```

```
<% var age
```

```
age = 18
```

```
%>
```

```
My age is <% =age %><BR>
```

```
</body>
```

```
</html>
```

דוגמא נוספת:

```
<%@ LANGUAGE=JScript %>
```

```
<html>
```

```
<head>
```

```
<title></title>
```

```
</head>
```

```
<body>
```

```
<% var x, y
```

```
x = 10
```

```
y = 15
```

```
x += y
```

```
%>
```

```
<% =x %>
```

```
</body>
```

```
</html>
```

בדוגמא הביטוי $x+=y$ זהה לביטוי $x = x + y$.
בצורה דומה $x=y$ שווה $x = x - y$.
וכך לגבי פעולות החשבון האחרות.

רשימת אופרטורים :

משמעות	סימן
הצבת ערך	=
השוואה	==
גדול מ	>
קטן מ	<
גדול או שווה ל	>=
קטן או שווה ל	<=
שונה מ	!=
ו (And) ... ו	&&
או (Or)	

הסתעפויות

שימוש ב if :

```
if (cops > 0 && crimes > 0)
    troubles = true
else
    happy = true
```

switch..case

פקודות אלו משמשות כדי לבדוק משתנה או ביטוי מול מספר אפשרויות.

דוגמא :

```
<%@ LANGUAGE=JScript %>
<html>

<head>
<title></title>
</head>

<body>
<% var code, message
code = "Let me in"
switch (code)
{
case "blah blah":
    message = "Hello there"
    break
case "Let me in":
    message = "Welcome home"
    break
default:
    message = "Access Denied"
```

```

}
%>
<% =message %>
</body>
</html>

```

אנו בודקים בכל פעם האם הערך של code הוא ערך מסויים. אם כן, אנחנו מדפיס הודעה מתאימה. אם הערך של code לא נמצא ברשימה שהגדרנו, אז מתבצעת ההוראה שבdefault.

לולאות for :

```

<%@ LANGUAGE=JScript %>
<html>
<%
var squared
for (var x=1; x <= 10; x++)
{
    squared = x * x
}
%>
<%=x%> squared is <%=squared%><p>
<% } %>
</HTML>

```

כל המספרים בין 1 ל 10 מוצגים עם החזקה הריבועית שלהם.

```

<%@ LANGUAGE=JScript %>
<html>

<head>
<title></title>
</head>

<body>
<%
var c
c = 0
while (c < 100) {
    c++
}
%>
<%=c%><BR>
<% } %>
</body>
</html>

```

דף זה יציג את כל המספרים בין 1 ל 100.

פונקציות

ניתן ליצור פונקציות על ידי הפקודה function

```
<%@ LANGUAGE=JScript %>
<html>

<head>
<title></title>
</head>

<body>
<% function square_it(v)
{
  var return_value
  return_value = v * v
  return return_value
}

var sq
sq=square_it(15)
%>
<%=sq%><BR>

</body>
</html>
```

הגדרנו פונקציה בשם square_it, המקבלת ערך v ומחזירה את v בריבוע, ולאחר מכן הדגמנו שימוש בפונקציה והעלתי את 15 בריבוע.

נדגים פונקציה של JScript – eval.
eval מקבלת מחרוזת שהיא תרגיל מתמטי ומחזירה את התוצאה שלו.
דוגמא:

```
<%@ LANGUAGE=JScript %>
<html>

<head>
<title></title>
</head>

<body>
<%
var formula, result
formula = "45 * (4 + 1)"
result = eval(formula)
%>
<%=formula %> is equal to <%=result %><BR>
</body>
</html>
```

בדף יהיה כתוב:

45 * (4 + 1) is equal to 225

כמו כן, הפונקציות המתמטיות השונות : $\tan()$, $\cos()$, $\sin()$ ועוד קיימות גם בJavaScript.

אובייקטים בASP

נתחיל כעת נושא חדש בASP – Objects.
 אובייקטים קיימים הן בVBScript והן בJScript.
 מספר מושגים:
 אובייקט – אוסף של משתנים ופונקציות תחת שם אחד.
 תכונה – משתנה בתוך אובייקט.
 שיטה (method) – פונקציה בתוך אובייקט.
 כדי לקבוע ערך של תכונה באובייקט, אנו משתמשים בsyntax הבא:

ערך = שם_תכונה.אובייקט

כדי לקרוא לשיטה, אנו משתמשים בsyntax הבא:

(ארגומנטים)שם_שיטה.אובייקט

כל מחרוזת בJScript היא למעשה אובייקט. למחרוזת יש תכונות ושיטות היכולות להיקרא כדי לנתח ולשנות את המחרוזת בדרך זו או אחרת.
 דוגמא:

```
<%@ LANGUAGE=JScript %>
<html>

<head>
<title></title>
</head>

<body>
<%
var message = "Hello World!!!"
%>
<%=message%><BR>
<%=message.toLowerCase()%><BR>
<%=message.toUpperCase()%><BR>
<%= "Welcome to my page".toUpperCase()%>
</body>
</html>
```

toLowerCase() היא שיטה ההופכת את כל המחרוזת לאותיות קטנות.
 toUpperCase() היא שיטה ההופכת את כל המחרוזת לאותיות גדולות.
 בדוגמא לעיל, הפלט יהיה:

```
Hello World!!!
hello world!!!
HELLO WORLD!!!
WELCOME TO MY PAGE
```

דוגמה לפונקציה נוספת :

```
<% var warning = "Death to all who enter here" %>
<% =warning.indexOf("enter", 0) %>
```

על הדף ייכתב 17, המיקום של המילה enter בתוך המחרוזת warning.

הפונקציה substr מאפשרת לנו לגזור חלקי מחרוזות ולהציגן או לבצע עליהן פעולות.
דוגמא :

```
<% var story = "Long time ago..." %>
<% =story.substr(10, 3) %>
```

בדף תופיע המילה ago. הפרמטר הראשון של הפונקציה מציין את התו הראשון ממנו מתחילים לגזור, והפרמטר השני מציין כמה תווים לגזור מהמחרוזת.

מספר פונקציות של מחרוזות מוסיפות תגים של HTML למסמך.
הנה רשימה של פונקציות אלו, והתגים אותן הן מציבות :

תג	פונקציה
	anchor(name)
	link(url)
<BIG>	big()
<SMALL>	small()
<BLINK>	blink()
	bold()
<I>	italic()
	fontcolor()
	fontsize()

דוגמאות :

```
<% var story = "Long time ago..." %>
<% =story.link("http://members.xoom.com/Underwar") %><BR>
<% =story.fontcolor("#ff44ff") %><BR>
```

אובייקטים המשותפים ל JScript ו VBScript

ישנם אובייקטים אליהם ניתן לפנות הן מ JScript והן מ VBScript. האובייקט הראשון שאציג הוא FileSystem. אובייקט זה נותן גישה לכוננים, מחיצות וקבצים על השרת. VBScript יוצרים אותו כך :

```
<%
Dim FileSystem
Set FileSystem = CreateObject("Scripting.FileSystemObject")
%>
```

: JScript ו

```
<%
var FileSystem
FileSystem = new ActiveXObject("Scripting.FileSystemObject")
%>
```

ל FileSystemObject עשרים שיטות.

הנה כמה מהן :

CopyFile משמש להעתיק קובץ ממקום אחד לשני.

MoveFile משמש להעביר קובץ ממקום אחד לשני.

CreateFolder יוצר מחיצה חדשה.

DeleteFile מוחק קובץ.

DeleteFolder מוחק ספרייה.

FileExists, FolderExists, DriveExists מאפשרים בדיקה האם קובץ מסוים, כונן

מסוים או מחיצה מסוימת קיימים.

GetFile, GetFolder, GetDrive מאפשרים לקבל את האובייקטים File, Drive, Folder

לפי בקשה.

דוגמא :

```
<%
Dim FileSystem
Set FileSystem = CreateObject("Scripting.FileSystemObject")
FileSystem.CopyFile "c:\stuff\*.*", "c:\backup\"
%>
```

ניתן להשתמש בפקודת For Each.. Next של VBScript כדי לעבור על אובייקטים באוסף אחד אחרי השני.

דוגמא

קבלת נתוני הכוננים בעזרת VBScript :

```

<html>

<head>
<title>Drives Info</title>
</head>
<body>
<%
  Dim FileSystem, drv, drvs
  Set FileSystem = CreateObject("Scripting.FileSystemObject")
  set drvs = FileSystem.Drives
  For each drv in drvs
%>
  Drive: <%=drv.DriveLetter%><P>
<% If drv.IsReady = True then %>
Volume: <%=drv.VolumeName %><P>
Space Free: <%=drv.FreeSpace%> bytes<P>
<% else %>
Drive is not ready. <P>
<% end if
next
%>
</body>
</html>

```

אותה דוגמא בJScript :

```

<%@ LANGUAGE=JScript %>
<html>

<head>
<title>Drives Info</title>
</head>
<body>
<%
var FileSystem, Drv, Drvs, EnumDrs
FileSystem = new ActiveXObject("Scripting.FileSystemObject")
Drs = FileSystem.Drives
EnumDrs = new Enumerator(Drs)
EnumDrs.moveFirst()
while (! EnumDrs.atEnd() ) { %>
Drive: <%=EnumDrs.item().DriveLetter %><P>
<% if (EnumDrs.item().IsReady == true) { %>
Volume: <% EnumDrs.item().VolumeName %><P>
Space Free: <% =EnumDrs.item().FreeSpace%> Bytes<P>
<% } else { %>
Drive is not ready.<P>
<% }

```

```

EnumDrvs.moveNext()
}
%>

</body>
</html>

```

השתמשנו באובייקט Enumerator. אובייקט זה מאפשר לנו להתעסק עם אוספים. כאשר אנו יוצרים אובייקט כזה, אנו שולחים לו אוסף. אז, בעזרת השיטות שלו, אנו עוברים בין האובייקטים באוסף:

- moveFirst() מעביר אותי לאובייקט הראשון.
- moveNext() מביא אותי לאובייקט הבא.
- atEnd() מחזיר אמת אם אני נמצא בסוף האוסף.
- item() מחזיר את האובייקט הנוכחי באוסף.

לאובייקט Drive מספר תכונות:

- DriveLetter מחזיר את האות המייצגת את הכונן.
- IsReady מחזיר true אם הכונן מוכן לגישה.
- FreeSpace מחזיר את מספר bytes הפנויים בכונן.
- TotalSpace מחזיר את מספר הבתים בכונן.
- FileSystem מחזיר את סוג מערכת הקבצים בכונן FAT, NTFS וכו'.
- RootFolder מחזיר את הספרייה הראשית בכונן. בעזרתה ניתן לגשת לכל שאר המחיצות בכונן.

לאובייקט Folder מספר שיטות ותכונות:

- השיטות הן Copy, Delete וmove המשמשות להעתקה, מחיקה והעברה של ספרייה. התכונות הן:
 - Name – שם הספרייה.
 - CreatedDate – תאריך יצירת הספרייה.
 - AccessedDate – תאריך כניסה אחרון.
 - ModifiedDate – תאריך שינוי אחרון.
 - Size – הגודל של כל הקבצים בספרייה.
 - Drive – הכונן בו נמצאת הספרייה.
 - ParentFolder – הספרייה הכוללת את הספרייה הנוכחית.
 - SubFolders – אוסף של רשימת הספריות תחת הספרייה הנוכחית.
 - IsRoot – הערך של תכונה זו אמת אם זאת הספרייה הראשית בכונן.
 - Files – אוסף של אובייקטים מסוג File הכולל את הקבצים בספרייה.

לאובייקט File מספר שיטות ותכונות:

- השיטות הן Copy, Delete וmove המשמשות להעתקה, מחיקה והעברה של קובץ. התכונות הן:
 - Name – שם הקובץ
 - CreatedDate – תאריך יצירת הקובץ.
 - AccessedDate – תאריך כניסה אחרון.
 - ModifiedDate – תאריך שינוי אחרון.
 - Size – הגודל של הקובץ.
 - Drive – הכונן בו נמצא הקובץ.
 - ParentFolder – הספרייה הכוללת את הקובץ.

דוגמא :

```

<html>
<head>
<title>Folders</title>
</head>
<body>
<%
  Dim FileSys, drv, rootdir, drivelet, fl
  drivelet = "C:"
  Set FileSys = CreateObject("Scripting.FileSystemObject")
  set drv = filesys.GetDrive(drivelet)
  set rootdir = drv.RootFolder
  for each fl in rootdir.SubFolders
%>
  C:\<%=fl.Name%><BR>
<%
  next
%>
</body>
</html>

```

דוגמא זו תציג את רשימת כל המחיצות בכוון C בדרך.

דוגמא בJScript :

```

<%@ LANGUAGE=JScript %>
<html>

<head>
<title>Files List</title>
</head>
<body>
<%
var FileSystem, drivelet, drv
var rootdir, rootfiles, enumfiles
FileSystem = new ActiveXObject("Scripting.FileSystemObject")
drivelet = "c:"
drv = FileSystem.GetDrive(drivelet)
rootdir = drv.RootFolder
rootfiles = rootdir.Files
enumfiles = new Enumerator(rootfiles)
enumfiles.moveFirst()
while (! enumfiles.atEnd() ) { %>
  <%=enumfiles.item().Name%><BR>
<%
  enumfiles.moveNext()
}
%>

```

```
</body>
</html>
```

דוגמא זו תציג את כל רשימת הקבצים שבספרייה הראשית.

ניתן ב ASP ליצור קבצים או להוסיף נתונים לקבצים קיימים.

```
<%
Dim filesystem, txtfile
Set filesystem = CreateObject("Scripting.FileSystemObject")
set txtfile = filesystem.CreateTextFile("c:\tmpfile.txt")
txtfile.WriteLine("Blah Blah Blah")
txtfile.Close
%>
```

בקטע קוד זה יצרנו קובץ בשם tmpfile.txt וכתבנו בתוכו Blah Blah Blah.
ישנן מספר שיטות בהן נוכל להשתמש:
Write – כתיבת מחרוזת לקובץ.
WriteLine – כתיבת מחרוזת לקובץ והוספת תו שורה חדשה.
WriteBlankLines מקבלת מספר כארגומנט וכותבת לקובץ מספר שורות ריקות
כמספר המצוין בארגומנט.
Close – סוגרת את הקובץ.

דוגמא נוספת:

```
<%
Dim filesystem, txtfile
Set filesystem = CreateObject("Scripting.FileSystemObject")
set txtfile = filesystem.OpenTextFile("c:\tmpfile.txt", 8, 0)
txtfile.WriteLine("P. S – Dabu Dabu Dabu")
txtfile.Close
%>
```

הארגומנט השני ב OpenTextFile מציין את מצב הפתיחה של הקובץ. 1 זה לפתוח לקריאה, 8 זה לפתוח להוספה.
הארגומנט השלישי קובע את סוג הפתיחה של הקובץ. 0 – ASCII. -1 – Unicode.
-2 – פותח לפי ברירת המחדל של המערכת.

דוגמא שלישית:

```
<%
Dim filesystem, txtfile, line
Set filesystem = CreateObject("Scripting.FileSystemObject")
set txtfile = filesystem.OpenTextFile("c:\tmpfile.txt", 1, 0)
do
  line = txtfile.ReadLine
%>
<%=line%>
%>
Loop Until txtfile.AtEndOfStream
txtfile.Close
%>
```

קטע קוד זה פותח את הקובץ לקריאה ושם את תוכנו בדף.

שיטות שניתן להשתמש בהן כאשר הקובץ נפתח לקריאה :
Read – מקבל מספר ומחזיר מספר תווים מהקובץ לפי המספר שצוין.
ReadLine – קורא שורה מהקובץ ומחזיר אותה כמחרוזת.
ReadAll – קורא את כל הקובץ ומחזיר אותו כמשתנה מחרוזת.
Skip – מקבל מספר תווים ומדלג עליהם בקובץ.
SkipLine – מדלג על שורה בקובץ.
Close – סוגר את הקובץ

תכונות בהן ניתן להשתמש :
AtEndOfStream – מכיל true אם הגענו לסוף הקובץ.
AtEndOfLine – מכיל true אם אנו נמצאים בסוף שורה.
Line – מחזיר את מספר השורה בקובץ.

טפסים

FORMS ב HTML מאפשרים למשתמש להכניס מידע לתוך פקדים כגון תיבות עריכה, כפתורי בחירה וכפתורי סימון, וכן לשלוח את המידע על ידי לחיצה על הכפתור SUBMIT.
 לדוגמא, נגדיר בדף FORM :

```
<form action="tst.asp" method=POST>
Name: <input name=Name type=edit><P>
</form>
```

ניצור דף שני בשם tst.asp בו נשים את הקוד הבא :

```
<html>
<head>
<title>Forms</title>
</head>
<body>
<h1>Here the information you entered:</h1>
<%=Request("Name")%><P>
</body>
</html>
```

Request("Name") זוהי בקשה למידע שהיה בפקד בשם Name.

ניתן ליצור דף אחד שישמש גם כקולט המידע וגם כמעבד המידע.
 דוגמא :

```
<html>

<head>
<title>Forms</title>
</head>
<body>
<%
If Request("Name")="" And Request("Address")="" Then %>
<h1>Please enter this information:</h1>
<form action="default.asp" method=POST>
Name: <input name=Name type=edit><P>
Address: <input name=Address type=edit><P>
<input type=submit value="Submit">
</form>
<% else %>
<h1>Here the information you entered:</h1>
<%=Request("Name")%><P>
<%=Request("Address")%><P>

<% end if %>

</body>
</html>
```

השורה <%= If Request("Name")="" And Request("Address")="" Then %> בודקת האם הוכנסו כבר הנתונים או לא. אם לא הוכנסו הנתונים, תופיע בקשה להכנסת הנתונים. אחרת יוצגו הנתונים בדף.

נדגים עכשיו כמה מאפשרויות האובייקט Response.

דוגמא אחת – בעזרת Response ניתן לכתוב מידע שיופיע בדף. בדוגמא הבאה נגדיר שגרה, בה ישנה הוראה לכתובת טקסט בדף, ואז נקרא לשגרה הזו.

```
<html>
<head>
<title>Welcome</title>
</head>
<body>
<%
Sub ShowHeader
Response.Write("<CENTER><H1>Welcome to my world</H1></CENTER>")
End Sub

ShowHeader
%>
</body>
</html>
```

אפשרות נוספת של Response היא לשלוח את הגולש לדף אחר בעזרת השגרה
: Redirect

```
<%
Response.Redirect("http://start.at/uw2000")
%>
```

דף בו תופיע שורה זו יעבור אוטומטית לאתר <http://start.at/uw2000>. הערה היא שכאשר קוראים שגרה זו, אין לשים תגים כגון HTML, HEAD או BODY לפניה במסמך. במידה ויהיו כאלו תגים, השרת יתעלם מההכוונה אל הדף החדש.

אובייקטי Application ואובייקטי Session.

אובייקטים אלו עובדים בצורה דומה ומשמשים למשימות דומות. באובייקט application כדי להחזיק מידע אודות תוכנות ASP ובאובייקט Session משתמשים כדי לעקוב אחרי כל משתמש בתוכנות ASP. כאשר יוצרים תוכנית ASP מתעורר לעיתים הצורך ליצור משתנים שיעברו מדף אל דף. לעזרתנו בא אובייקט Session. כדי ליצור אובייקט Session משתמשים בצורת הכתיבה הבאה :

ערך = Session("שם_משתנה")

דוגמא :

Session("Total_Cost") = 1500
 בדוגמא זו יצרתי משתנה בשם Total_Cost והצבתי לו את הערך 1500.
 ניתן ליצור אף משתנים גלובליים מסוג Application, בצורה הבאה :

ערך = Application("שם_משתנה")

אולם, יש לזכור שמשנתני Application הם גלובליים, ומספר משתמשים יכולים לשנותם בו זמנית. על מנת לגרום שרק משתמש אחד יוכל לשנות משתנה זה ניתן להשתמש בשגרות Lock וUnlock של אובייקט Application. דוגמא :

Application.Lock
 Application("Number_of_visitors") = 10
 Application.Unlock

דוגמא לcounter פשוט שנכתב בעזרת משתנה Application :

```
<HTML>
<HEAD>
<TITLE>Application Variable</TITLE>
</HEAD>
<BODY>
<%
Application.Lock
Application("Counter1") = Application("Counter1") + 1
Application.Unlock
%>
Visitors on this page: <%=Application("Counter1")%>
</BODY>
</HTML>
```

הערה היא שהמשתנה Counter1 יעלם עם סיום התוכנית. אם רוצים ליצור Counter אמיתי יש לשמור את מספר המבקרים בתוך קובץ ולקרוא את המספר מהקובץ.

ההבדל בין משתני Session לבין משתני Application הוא זה :
 אם עשרה אנשים משתמשים בתוכנית ASP שלך בו זמנית באותו זמן, עשרה עותקים של משתנה Session ייוצרו, אולם ייווצר רק עותק אחד של משתנה Application.

נניח שיצרת פונקציה שאתה רוצה שתהיה במספר דפים אצלך.
 דרך אחת היא לכתוב את הפונקציה בכל אחד מהדפים.
 דרך שנייה היא להציב את השורה הבאה בראש הדף שאליו אתה רוצה להוסיף את הפונקציה.

```
<!--#include file="שם_הקובץ_בו_כתובה_הפונקציה"-->
```

נביא עכשיו דוגמא שתדגים את הדברים אותם ראינו במסמך זה.
 בדוגמא זו ניבנה Guest Book לאתר. ה Guest Book מורכב ממספר דפים.
 הדף הראשון הוא default.asp :

```
<html>
<head>
<title>Welcome</title>
</head>
<body>
Sign my <A HREF="gb.asp">Guestbook</A><BR><BR>
<%
Application.Lock
Application("Guest_Counter") = Application("Guest_Counter") + 1
Application.Unlock
%>
Visitors on this page: <%=Application("Guest_Counter")%>

</body>
</html>
```

בדף זה יש קישור כדי לחתום ב Guest Book, וכן יש Counter.
 הדף הבא הוא gb.asp, בו ממלא המבקר את פרטיו :

```
<html>
<head>
<title>Guest Book</title>
</head>
<body bgcolor="WHITE">
Welcome to my guest book.<BR>
<form method="POST" action="gb2.asp">
First Name: <input type="text" size=20" name="FirstName"><P>
Last Name: <input type="text" size=20" name="LastName"><P>
Gender: <INPUT TYPE="radio" checked Name="Gender" Value="Male">Male
<INPUT TYPE="radio" Name="Gender" Value="Female">Female<P>
Age: <input type="text" size="10" name="Age"><P>
<input type="submit" name="Submit" value="Thank you!"><P>
</Form>
</body>
</html>
```

זהו דף HTML פשוט בו יש Form שמקבל נתונים מהמשתמש ומשגר אותם אל הדף
 gb2.asp.

הדף gb2.asp :

```

<html>
<head>
<title>Guest Book</title>
</head>
<body>
First Name: <%=Request("FirstName")%><BR>
Last Name: <%=Request("LastName")%><BR>
Gender: <%=Request("Gender")%><BR>
Age: <%=Request("Age")%><BR>
<%
Dim filesystem, peoplefile
set filesystem=CreateObject("scripting.FileSystemObject")
Set peoplefile = filesystem.OpenTextFile("d:\inetpub\wwwroot\Guest\gbpeople.txt", 8,
true)
peoplefile.WriteLine request("FirstName")
peoplefile.WriteLine request("LastName")
peoplefile.WriteLine request("Gender")
peoplefile.WriteLine request("Age")
%>
<BR>
The information has been saved.<BR>
<A HREF="gblist.asp">See others who signed it</A>
</body>
</html>

```

בדף זה מוצגת האינפורמציה ששלח המשתמש, ולאחר מכן, באמצעות אובייקט
 FileSystemObject נשמרים הנתונים לתוך קובץ.

הדף האחרון הוא gblist.asp בו מוצגים האנשים שחתמו ב-Guest Book :

```

<HTML>
<HEAD>
<TITLE>Guest Book List</TITLE>
</HEAD>
<BODY>
<H1>Guest Book List</H1>
<% Dim filesystem, peoplefile, fname, lname, gender, age
set filesystem=CreateObject("Scripting.FileSystemObject")
set peoplefile = filesystem.OpenTextFile("d:\inetpub\wwwroot\guest\gbpeople.txt", 1)
Do while not peoplefile.atEndOfStream
  fname=peoplefile.ReadLine
  lname=peoplefile.ReadLine
  gender=peoplefile.ReadLine
  age=peoplefile.ReadLine
  Response.Write "First Name: <B>" & fname & "</b><BR>"
  Response.Write "Last Name: <B>" & lname & "</b><BR>"
  Response.Write "Gender: <B>" & gender & "</b><BR>"
  Response.Write "Age: <B>" & age & "</b><P>"
loop

```

```
peoplefile.close
%>
</BODY>
</HTML>
```

גם פה אנו יוצרים אובייקט `FileSystemObject`, ואז אנו פותחים את קובץ הטקסט בו שמורים הנתונים ובעזרת לולאת `Do while` אנו קוראים את כל הנתונים עד שאנו מגיעים לסוף הקובץ.

דוגמא

דוגמא לתוכנית שתשגר את המבקר בו כל פעם לאתר אקראי כלשהו :

```
<%
dim sites(10), num
Randomize

Sites(1)="http://www.borland.com"
Sites(2)="http://www.netscape.com"
Sites(3)="http://www.rootshell.com"
Sites(4)="http://start.at/uw2000"
Sites(5)="http://start.at/jaguar2000"
Sites(6)="http://www.developer.com"
Sites(7)="http://linuxberg.com/"
Sites(8)="http://www.freshmeat.net/"
Sites(9)="http://www.xoom.com/"
Sites(10)="http://java.sun.com/"

num = int(rnd*10)+1
Response.Redirect sites(num)
%>
```

בתוכנית זאת לא שמנו תגי HTML, אלא רק את הוראת ה-`Redirect`. בתוכנית יצרנו מערך של 10 משתנים המכילים כתובות אתרים, ואז הגרלנו מספר רנדומלי ושיגרנו את המשתמש אל האתר בעל המספר האקראי שנבחר.

דוגמא

קודם הצגנו Counter שפועל באמצעות משתנה Application. עכשיו נדגים counter שפועל באמצעות קובץ טקסט בו נשמר מספר המבקרים:

```
<HTML>
<HEAD>
<TITLE>Counter</TITLE>
</HEAD>
<BODY>
<%
Dim FileSystem, CounterFile, Visitors
Set FileSystem = CreateObject("Scripting.FileSystemObject")
%>
<%
if FileSystem.FileExists(Server.MapPath("/SiteCounter.txt")) = False then
    set CounterFile = FileSystem.CreateTextFile(Server.MapPath("/SiteCounter.txt"))
    CounterFile.WriteLine("1")
    CounterFile.Close
End If

if FileSystem.FileExists(Server.MapPath("/SiteCounter.txt")) = True then
    set CounterFile = FileSystem.OpenTextFile(Server.MapPath("/SiteCounter.txt"), 1,
0)
    Visitors = CounterFile.ReadLine
    CounterFile.Close
    set CounterFile = FileSystem.CreateTextFile(Server.MapPath("/SiteCounter.txt"))
    CounterFile.WriteLine(Visitors + 1)
    CounterFile.Close
End If
%>
Visitors: <%=Visitors%>

</BODY>
</HTML>
```

דוגמא זו משלבת רבים מהדברים אותם הסברנו וכן שגרה חדשה.

```
if FileSystem.FileExists(Server.MapPath("/SiteCounter.txt")) = False then
```

שורה זו היא שורת מפתח בתוכנית.

Server.MapPath זוהי שיגרה המחזירה את הספרייה בה נמצא האתר, למשל,
rootlocation = Server.MapPath("/")

במשתנה rootlocation יוצב הערך "C:\inetpub\wwwroot".

בשורה זו אנו בודקים האם קיים במחיצה של האתר קובץ בשם SiteCounter.txt. אם הוא לא קיים, אנו יוצרים אותו ושמים בו את המספר 1 (מבקר אחד). אם הוא קיים, אנו פותחים אותו, קורא את מספר המבקרים, ואז דורסים אותו בקובץ חדש

באותו שם, בו אנו שמים את מספר המבקרים + 1. לאחר מכן אנו מציגים את מספר המבקרים.

רכיבי שרת

כעת נעבור לנושא חדש : רכיבי שרת (ActiveX Server Components).
רכיבי שרת הם תוכנות קטנות הכתובות ב-C++, Visual Basic או שפות אחרות, המשתמשות ב-ASP כדי לבצע פעולות שונות. על מנת להשתמש ברכיב שרת עליך ליצור עותק שלו ולהשתמש בו.
IIS בא ביחד עם מספר רכיבי שרת שניתן להשתמש בהם באופן מיידי. נציג אחד מהם :

סוג דפדפן

אובייקט "סוג דפדפן" (Browser Type). בעזרת אובייקט זה אתה יכול לקבל מידע על הדפדפן בו משתמש המשתמש, ולשלוח קוד מתאים שיתאים ליכולות השרת.
דבר ראשון עלינו ליצור עותק של אובייקט סוג הדפדפן :

```
set browser = Server.CreateObject("MSWC.BrowserType")
```

לאחר מכן ניתן להשתמש במשתנה browser כדי לקבל מידע על הדפדפן של המשתמש. לדוגמה, browser.Tables יחזיר true אם הדפדפן תומך בטבלאות וfalse אם לא.

דוגמה הכוללת שימוש ברכיב סוג הדפדפן :

```
<HTML>
<HEAD>
<TITLE>BrowserType Object</TITLE>
</HEAD>
<BODY>
<% set browser = Server.CreateObject("MSWC.BrowserType") %>
Browser Version: <%=browser.Version %><BR>
<% if browser.Frames = True then %>
Your browser supports frames<BR>
<% else %>
Your browser doesn't supports frames<BR>
<% end if %>
<BR>
<% if browser.Tables = True then %>
<TABLE BORDER = 1>
<TR>
<TD>1, 1</TD>
<TD>1, 2</TD>
</TR>
<TR>
<TD>2, 1</TD>
<TD>2, 2</TD>
</TR>
```

```

</TR>
</TABLE>

<% else %>
<pre>
1,1 1,2
2,1 2,2
</pre>
<BR>
<% end if %>

</BODY>
</HTML>

```

בתחילה השורה
<% =browser.Version %> Browser Version: גורמת להצגת גרסת הדפדפן על המסך.

לאחר מכן בעזרת browser.Frames אני בודק אם הדפדפן תומך במסגרות, ומציג הודעה מתאימה, ולאחר מכן אני בודק אם הדפדפן תומך בטבלאות. אם כן אני מציג טבלה, אחרת אני מציג טקסט פשוט. דוגמא זו מדגימה יצירת דף אחד שמתאים לדפדפנים רבים. הנה רשימת התכונות של רכיב סוג הדפדפן:

- Browser – מכיל את שם הדפדפן
- Version – מכיל את גרסת הדפדפן
- Frames – האם הדפדפן תומך במסגרות או לא
- Tables – האם הדפדפן תומך בטבלאות או לא
- BackgroundSounds – האם הדפדפן תומך ב<Bgsound>
- ActiveXControls – האם הדפדפן תומך בActiveX Controls.
- JavaApplets – האם הדפדפן תומך בJava Applets.
- VBScript – האם הדפדפן תומך בVBScripts.
- JavaScript – האם הדפדפן תומך בJavaScript.
- Cookies – האם הדפדפן תומך בcookies.

EOF